UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

The Lanczos-preconditioned, folded (sin x)/x interpolator: discussion and desktop-computer program in BASIC +

by

W. P. Hasbrouck[1] and L. F. Bailey[2]

Open-File Report 84-266

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of brand names in this report is for descriptive purposes only and does not imply endorsement by the USGS. Although the program in this report has been extensively tested, the USGS makes no guarantee whatever of correct results.

[1] Golden, Colorado

[2] Retired from the National Oceanic and Atmospheric Administration, Boulder, Colorado

+ Presented in part at the 38th Annual International Society of Exploration Geophyicist meeting, October 2, 1968.

1984

ABSTRACT

The Lanczos-preconditioned, folded (sin x)/x interpolator produces a selected number of interpolated values between sequential pairs of equally spaced data without changing the values in the original data set. Best results are obtained when the original data appear to be composed of a group of trigonometric functions. This report presents a development of the interpolator and lists a computer program for its implementation. The program was developed for use on a desktop, stand-alone computer and was written in the extended BASIC language used by the Tektronix 4051 Graphic System.

INTRODUCTION

The Lanczos-preconditioned, folded (sin x)/x interpolator (termed the "interpolator" for brevity in this report) is an interpolation procedure designed to operate on data that are equally spaced, meet the Nyquist criterion, are representable by numerical sequences that reasonably can be fitted by continuous functions satisfying the Dirichlet conditions, and have the appearance of being composed of a set of trigonometric functions (Bailey, 1966). When applied to a summed set of periodic trigonometric functions, the interpolator will give exact results, as will the Fourier methods upon which the interpolator is based. When applied to field data for which non-zero values and slopes occur at the end points, the interpolator produces results that look realistic.

Significant trends with first-order discontinuites are removed in the program by first linearly detrending and then reflecting the detrended function using odd-function symmetry (Lanczos, 1956 p. 241). In this way, the continuity of the function and its first derivatives at the termini of the expanded data interval is guaranteed, and satisfactory convergence of the trigonometric series is assured. We call this prior treatment of the data "Lanczos preconditioning." One can think of this conditioning process as one of forcing the original data set into being a well-behaved periodic function.

After the properly sampled function is Lanczos preconditioned, it is operated upon by the modified kernel of Zygmund (1952, p. 50) as discussed by Hamming (1962, p. 282).

The computer program for implementation of the interpolator was written in an extended BASIC language developed by Tektronix, Inc. for use with their 4051 Graphic System. Three pieces of Tektronix computing equipment are required by the program: a 4051 Graphic System with a 32K-byte memory, a 4924 Digital Tape Cartridge, and a 4631 Hard Copy unit. The program is self prompting. In working through the sample problem, the user will notice that the program prints questions and requests followed by a flashing question mark. The computer then waits for a response from the keyboard. Replies entered in running the sample problem are enclosed in boxes.

DEVELOPMENT OF THE INTERPOLATOR

A continuous band-limited signal, when sampled at the Nyquist interval, can be completely described by the well-known Shannon sampling theorem:

$$f(t) = \sum_{n=-\infty}^{\infty} f(nT) \frac{\sin \omega_m (t-nT)}{\omega_m (t-nT)} , \qquad (1)$$

where $\omega_m = 2 \pi f_m$, and $T = 1/(2f_m)$. Because any function of finite duration can be thought of as the product of itself and a rectangular pulse whose elevated section is of the same duration, the Fourier transform of any finite-length function will contain spectral components above the Nyquist frequency. However, if the original function is Lanczos preconditioned and then forced into periodicity at this new period, then the infinite trigonometric series of (1) can be expressed as a finite series. Development of the folded (sin x)/x coefficients proceeds on this basis. Systematic summation of the cross products of these coefficients with the ordinal values of the sampled function yields the detrended interpolated value.

Consider, as shown in figure 1, a finite-length function sampled at eight points at an interval of T. Shown beneath the plot of the function are the associated Lanczos preconditioned function and some of its periodic repetitions. The interpolated value at the point P is desired. In this example, the number of independent sample points N is 7, and the length of the Lanczos preconditioned interval L is 14T. Using d as the distance from the origin at M = 0 to the point P, adopting the index K to denote the number of repeated intervals, and letting p be a decimal expressing the proportional position of the interpolated point (for example, p = 0.2 would mean point P is a fifth of T), then the (sinx)/x weighting factors associated with each ordinal value would be given by

$$\sum_{K=-\infty}^{\infty} \frac{\sin(p-M)\pi}{(p-M-2NK)\pi} . \qquad (2)$$

Thus for the example shown on figure 1,

$$f(p) = \sum_{M=1}^{6} B(M) \sum_{K=-\infty}^{\infty} \left[\frac{\sin(p-M)\pi}{(p-M-2NK)\pi} - \frac{\sin(p+M)\pi}{(p+M-2NK)\pi} \right] , \qquad (3)$$

which can be expressed as

$$f(p) = \sum_{M=1}^{N-1} B(M) \left[A(M) - A(-M) \right] , \qquad (4)$$

where A(M) is the folded (sinx)/x coefficient, given by



Figure 1. Finite-length function sampled at N=7 independent points at interval T and its repetitive Lanczos preconditioned representation at period L=14. The ordinal values of each sample is b(M) on the original function and B(M) on the periodic function, the index ranging from -6 to 7. Distance d is from the origin at M=0 to the point P at which the interpolation is desired. An index K is assigned to identify each repeated periodic interval. Shown by dimensioned distances are the distances in specific and generalized terms for the point P to ordinates B(3) and B(-3) for K=-1 to K=1.

$$A(M) = \sum_{K=-\infty}^{\infty} \frac{\sin(p-M)\pi}{(p-M-2NK)\pi} \quad . \tag{5}$$

For an interpolated point lying between positions n and n+1, the detrended interpolated value would be obtained from the summation

$$f(p) = \sum_{M=1}^{N-1} B(M) \, [A(M-n) - A(-M-n)] \; , \tag{6}$$

where if (-M-n) is less than -(N-1), then (-M-n) is replaced by the term (-M-n+2N). As an example, consider the point P to be between the 4th and 5th points. The interpolated value would be with n = 4,

$$\begin{aligned} f(p) = {} & B(1)\ [A(-3) - A(-5)] + B(2)\ [A(-2) - A(-6)] + \\ & B(3)\ [A(-1) - A(+7)] + B(4)\ [A(+0) - A(+6)] + \\ & B(5)\ [A(+1) - A(+5)] = B(6)\ [A(+2) - A(+4)]. \end{aligned} \tag{7}$$

Thus once the coefficients are determined, 2N being required for each p value, it is a simple and quick computer task to find f(p).

The development of the coefficients from their infinite series representation follows: With L = 2N and letting m = p-M, (5) becomes

$$A(M) = \frac{\sin m\pi}{\pi} \sum_{K=-\infty}^{\infty} \frac{1}{m - LK} \quad , \tag{8}$$

which can also be written

$$A(M) = \frac{\sin m\pi}{m\pi} + \frac{2m}{\pi} \sin m\pi \sum_{K=1}^{\infty} \frac{1}{m^2 - L^2K^2} \quad . \tag{9}$$

Since

$$\cot x = \frac{1}{x} + 2x \sum_{K=1}^{\infty} \frac{1}{x^2 - K^2\pi^2} \quad , \tag{10}$$

then

$$\sum_{K=1}^{\infty} \frac{1}{m^2 \quad -L^2 K^2} = \frac{\pi}{2Lm} \left(\cot \frac{m\pi}{L} - \frac{L}{m\pi}\right) . \tag{11}$$

Upon substitution of (11) into (9),

$$A(M) = \frac{\sin m\pi}{L} \cdot \cot \frac{m\pi}{L} , \tag{12}$$

and with resubstitution of the m and L quantities, (12) becomes

$$A(M) = \frac{\sin(p-M)\pi \cos \frac{(p-M)\pi}{2N}}{2N \sin \frac{(p-M)\pi}{2N}} . \tag{13}$$

Interpolation is completed upon reinstatement of the portion of ordinal values removed during linear detrending.

EXAMPLES USING THE COMPUTER PROGRAM

The program of this report was designed as a complete program to compute interpolated values within a specific traverse or time series. In its more common application, the interpolator is included as a subroutine of a larger program. It is our hope that after study of the development in the previous section and after working through the sample problems that the reader will be sufficiently familiar with the method to adapt it to his particular needs.

As with most data-processing programs, this one begins with a data entry section. The first question you are asked (fig. 2a) is: DO YOU WANT TO INPUT VALUES FROM A TAPE? (Y OR N). If you answer with a Y, you are prompted first to insert a data tape in the Tektronix 4924--the Digital Tape Cartridge--and then to enter the number of the file in which the data are stored.

If you answer the first question with an N, then you are asked: DO YOU WANT TO WORK WITH PROFILE DATA? (Y OR N). If a Y is your reply, then a display such as shown on figure 2a will develop; if an N is your entry, then the program assumes you want to work with time-series data.

After making entries as prompted, you are asked as to whether you want to store the data you have entered. On figure 2a, storage was selected; on figure 2b, it was not. If you elect to store the keyboard-entered data on a master data tape (MDT), you are requested to supply the number of the file in which the data are to be stored. Upon receiving a Y response to the question of sufficient file length, data are stored on tape.

Next you are asked to enter the number of interpolation points between adjacent values. In both examples we entered a 9. Thus, for the profile example (fig. 2a), we wanted interpolated values at a spacing of 1 m along the traverse; for the time-series example (fig. 2b), we wanted interpolated values at a tenth of a msec.

```
LANCZOS-PRECONDITIONED FOLDED SINX/X INTERPOLATION

DO YOU WANT TO INPUT VALUES FROM A TAPE? (Y OR N) [N]

DO YOU WANT TO WORK WITH PROFILE DATA? (Y OR N) [Y]
   Station spacing = 10
Units, 12 char max = METERS
Start profile dist = 30
  End profile dist = 100
  Number of values = 8
DATA IDENTIFICATION (max of 22 characters):      TEST: OPEN-FILE REPORT
STA NO.   STA DIST    STA VALUE
  1          30          65
  2          40          17
  3          50          31
  4          60          63
  5          70          60
  6          80          26
  7          90          20
  8         100          10

DO YOU WANT TO STORE THESE VALUES ON A MDT? (Y OR N) [Y]
INSERT MDT IN 4924     FILE NO. = [53]
LENGTH OF FILE REQUIRED = 256
IS FILE LENGTH SUFFICIENT? (Y OR N) [Y]
DATA ARE STORED AND RETRIEVABLE FROM FILE 53

NO. OF INTERPOLATION POINTS BETWEEN ADJACENT VALUES = [9]

DO YOU WANT TO PLOT ORIGINAL FUNCTION? (Y OR N) [Y]
```

Figure 2a. Copy of screen display showing example of data entered from the keyboard for the profile-data sample problem.

```
LANCZOS-PRECONDITIONED FOLDED SINX/X INTERPOLATION

DO YOU WANT TO INPUT VALUES FROM A TAPE? (Y OR N) [N]

DO YOU WANT TO WORK WITH PROFILE DATA? (Y OR N) [N]

YOU HAVE ELECTED TO WORK WITH TIME-SERIES DATA
   Sample interval = 1
Units, 12 char max = MSEC
        Start time = 30
          End time = 37
  Number of values = 8
DATA IDENTIFICATION (max of 22 characters):      TEST: OPEN-FILE REPORT
DATA PT.    TIME       VALUE
  1          30          65
  2          31          17
  3          32          31
  4          33          63
  5          34          60
  6          35          26
  7          36          20
  8          37          10

DO YOU WANT TO STORE THESE VALUES ON A MDT? (Y OR N) [N]

NO. OF INTERPOLATION POINTS BETWEEN ADJACENT VALUES = [9]

DO YOU WANT TO PLOT ORIGINAL FUNCTION? (Y OR N) [Y]
```

Figure 2b. Copy of screen display showing example of data entered from the keyboard for the time-series sample problem.

Finally you are asked if you want to plot the original function. Our general practice is to request this plot (fig. 3) and then scan it for patent errors. The title block of this plot, and all others of the program, contains: an identification of the function (first line), the number of values and the minimum and maximum values, and where they occur, (second line), and a short description of the original function (third line). The mean value of the function is drawn across the plot.



Figure 3. Copy of screen plot showing the original function.

Upon making or skipping the plot of the original function, you are asked if you want to see a plot of the detrended function (fig. 4). At this place in the procedure, end points of the function equal zero.

Figure 5 shows the original function after it has been Lanczos preconditioned; that is, detrended and odd-function folded. It is this function that is forced into periodicity. Note that at the termini the derivatives are continuous and equal.

In order to save processing time, figures 4 and 5 usually are not displayed; however, in this report, the plots are included in order to make the computing procedure of the program easier to follow.

Figure 6 is a tabulation of the folded (sin x)/x coefficients. Note in the listing of coefficients that the number of sets of coefficients equals the number of values in the Lanczos preconditioned

TEST: OPEN-FILE REPORT   DETRENDED ORIGINAL FUNCTION
NO. OF VALUES=8   MIN=-40.1428571429 @ 40   MAX=26.4285714286 @ 70
ORIGINAL PROFILE STARTS AT 30 AND ENDS AT 100 WITH STA SPA = 10 METERS



Figure 4. Copy of screen plot showing the detrended function.

TEST: OPEN-FILE REPORT   FUNCTION AFTER LANCZOS PRECONDITIONING
NO. OF VALUES=15   MIN=-40.1428571429 @ 40   MAX=40.1428571429 @ 160
ORIGINAL PROFILE STARTS AT 30 AND ENDS AT 100 WITH STA SPA = 10 METERS



Figure 5. Screen plot of the Lanczos preconditioned function.

function--excluding the zero values at the termini, and that the number of coefficients in each set equals the number of interpolated values between adjacent values of the original function. Thus for this sample problem, 126 (14x9) coefficients must be computed.

In repeated calculations with the interpolator, if the number of original data points and the number of interpolated values are to be the same; for example, when constructing a map consisting of profiles of equal number of stations; then, the coefficients need be computed only once. These coefficients are then stored for later recall.

Figure 7 shows the interpolated function. An overlay of figure 7 on figure 3 (the plot of the original data) and a comparison of the listing of the values of the interpolated function (fig. 8) to the data as entered (fig. 2a) would show that values of the original data set have been preserved.

An example of the procedure to be followed in order to store the interpolated function is displayed on the bottom five lines of figure 8.

A plot of the interpolated function for the time-series example is shown on figure 9 followed by a listing of values of this function on figure 10. If this sample problem represented a section of a seismic record taken at a sample interval of 1 msec, but you needed to read arrival times to the nearest 0.1 msec, then the interpolator could be used to produce a function that subsequently could be searched for trough, peak, and zero-crossing times.

ADDITIONAL COMMENTS ON THE PROGRAM

In order to put the programs of this report to work, you must know how to perform the following operations:

1. transcribe the programs into the computer,
2. store the programs on magnetic tape,
3. retrieve the programs from magnetic tape,
4. enter information from the keyboard, and
5. copy the screen display.

Four control characters (ones requiring the holding down of the control key as the letter is entered) are used in the programs: G (ring bell), K (move cursor up one line), L (erase screen and move cursor to the HOME position), and the RUB OUT (move cursor to the left margin and down one line). In the printed listing these control characters are shown as G_, K_, L_, and __, respectively.

REFERENCES

Bailey, L. F., 1966, Tables of folded-sin x/x interpolation coefficients: U.S. Coast and Geodetic Survey Research Paper (Supt. of Doc., U.S. Government Printing Office, Washington, D. C. 20402).

Hamming, R. W., 1962, Numerical methods for scientists and engineers: McGraw-Hill, Inc., New York, 411 p.

Lanczos, Cornelius, 1956, Applied analysis: Englewood Cliffs, N. J., Prentice Hall, Inc., 539 p.

Zygmund, Antoni, 1952, Trigonometric series: second ed, (vol. 1), Cambridge University Press, New York, 329 p.

```
FOLDED (SIN X)/X COEFFICIENTS FOR TEST: OPEN-FILE REPORT

 0.00451939332358  0.0076190926286   0.00915255027622  0.00920209703897
 0.00804806713572  0.00611404540995  0.00389609433585  0.00188553329418
 4.95392093E-4

 -0.010025869444   -0.0179451052629  -0.0231800015755  -0.0254955343384
 -0.0249939381391  -0.0220726424554  -0.0173526155096  -0.0115870301623
 -0.00556181984929

 0.0168061519212   0.0305036416518   0.0400348861437   0.0448419221339
 0.0448815460977   0.0405878439955   0.0327894391345   0.0225929094288
 0.0112465730686

 -0.0264387464851  -0.0480553110351  -0.0632214193218  -0.0710519378273
 -0.0714285714286  -0.0649502239135  -0.0528195839813  -0.0366808935594
 -0.0184275584032

 0.0433199999671   0.078020573428    0.101841603743    0.113700034717
 0.113677920213    0.102913502628    0.0834104799323   0.0577869281697
 0.0289894764279

 -0.0875975055223  -0.152128002116   -0.192439523911   -0.209075070223
 -0.204131129234   -0.181005787528   -0.144060778271   -0.0982279971737
 -0.048594443373

 0.983466534561    0.934861110374    0.857096563745    0.754793097195
 0.633946105355    0.501498653168    0.36485230526     0.231354550334
 0.107802421714

 0.107802421714    0.231354550334    0.364852305259    0.501498653168
 0.633946105355    0.754793097195    0.857096563745    0.934861110374
 0.983466534561

 -0.048594443373   -0.0982279971737  -0.144060778271   -0.181005787528
 -0.204131129234   -0.209075070223   -0.192439523911   -0.152128002116
 -0.0875975055223

 0.0289894764279   0.0577869281696   0.0834104799323   0.102913502628
 0.113677920213    0.113700034717    0.101841603743    0.078020573428
 0.0433199999671

 -0.0184275584032  -0.0366808935594  -0.0528195839813  -0.0649502239135
 -0.0714285714286  -0.0710519378273  -0.0632214193218  -0.0480553110351
 -0.0264387464851

 0.0112465730686   0.0225929094288   0.0327894391345   0.0405878439955
 0.0448815460977   0.0448419221339   0.0400348861437   0.0305036416518
 0.0168061519212

 -0.00556181984929 -0.0115870301623  -0.0173526155096  -0.0220726424554
 -0.0249939381391  -0.0254955343384  -0.0231800015755  -0.0179451052629
 -0.010025869444

 4.95392093E-4     0.00188553329418  0.00389609433585  0.00611404540995
 0.00804806713572  0.00920209703897  0.00915255027622  0.0076190926286
 0.00451939332358

DO YOU WANT TO PLOT INTERPOLATED FUNCTION? (Y OR N) Y
```

Figure 6. Tabulation of the folded (sin x)/x coefficients.

TEST: OPEN-FILE REPORT INTERPOLATED FUNCTION WITH 63 ADDITIONAL VALUES
NO. OF VALUES=71 MIN=10 @ 100 MAX=68.1533130578 @ 65
ORIGINAL PROFILE STARTS AT 30 AND ENDS AT 100 WITH STA SPA = 10 METERS



Figure 7. Copy of screen plot showing the interpolated function for the profile-data sample problem.

TEST: OPEN-FILE REPORT INTERPOLATED FUNCTION WITH 63 ADDITIONAL VALUES
NO. OF VALUES=71 MIN=10 @ 100 MAX=68.1533130578 @ 65
ORIGINAL PROFILE STARTS AT 30 AND ENDS AT 100 WITH STA SPA = 10 METERS

STATION	VALUE	STATION	VALUE	STATION	VALUE
30.0	65.0	31.0	58.3	32.0	51.7
33.0	45.4	34.0	39.5	35.0	34.1
36.0	29.3	37.0	25.1	38.0	21.7
39.0	19.0	40.0	17.0	41.0	15.8
42.0	15.3	43.0	15.5	44.0	16.4
45.0	17.8	46.0	19.7	47.0	22.0
48.0	24.8	49.0	27.8	50.0	31.0
51.0	34.4	52.0	37.9	53.0	41.4
54.0	45.0	55.0	48.5	56.0	51.8
57.0	55.0	58.0	58.0	59.0	60.6
60.0	63.0	61.0	65.0	62.0	66.5
63.0	67.6	64.0	68.1	65.0	68.2
66.0	67.6	67.0	66.5	68.0	64.8
69.0	62.7	70.0	60.0	71.0	56.9
72.0	53.5	73.0	49.9	74.0	46.1
75.0	42.3	76.0	38.5	77.0	34.9
78.0	31.6	79.0	28.6	80.0	26.0
81.0	23.8	82.0	22.1	83.0	20.8
84.0	20.0	85.0	19.5	86.0	19.3
87.0	19.4	88.0	19.6	89.0	19.8
90.0	20.0	91.0	20.1	92.0	20.0
93.0	19.6	94.0	19.0	95.0	18.1
96.0	16.9	97.0	15.4	98.0	13.7
99.0	11.9	100.0	10.0		

DO YOU WANT TO STORE INTERPOLATED FUNCTION? (Y OR N) Y
INSERT MDT IN 4924 FILE NO. = 54
LENGTH OF FILE REQUIRED = 1024
IS FILE LENGTH SUFFICIENT? (Y OR N) Y
DATA ARE STORED AND RETRIEVABLE FROM FILE 54

Figure 8. Listing of values of the interpolated function for the profile sample problem and five-line procedure for storing results.

TEST: OPEN-FILE REPORT INTERPOLATED FUNCTION WITH 63 ADDITIONAL VALUES
NO. OF VALUES=71   MIN=10 @ 37   MAX=68.1533130578 @ 33.5
ORIGINAL TIME BEGINS AT 30 AND ENDS AT 37 WITH SAMP INT = 1 MSEC



Figure 9. Copy of screen plot showing interpolated function for the time-series sample problem.

TEST: OPEN-FILE REPORT INTERPOLATED FUNCTION WITH 63 ADDITIONAL VALUES
NO. OF VALUES=71   MIN=10 @ 37   MAX=68.1533130578 @ 33.5
ORIGINAL TIME BEGINS AT 30 AND ENDS AT 37 WITH SAMP INT = 1 MSEC

TIME	VALUE	TIME	VALUE	TIME	VALUE
30.0	65.0	30.1	58.3	30.2	51.7
30.3	45.4	30.4	39.5	30.5	34.1
30.6	29.3	30.7	25.1	30.8	21.7
30.9	19.0	31.0	17.0	31.1	15.8
31.2	15.3	31.3	15.5	31.4	16.4
31.5	17.8	31.6	19.7	31.7	22.0
31.8	24.8	31.9	27.8	32.0	31.0
32.1	34.4	32.2	37.9	32.3	41.4
32.4	45.0	32.5	48.5	32.6	51.8
32.7	55.0	32.8	58.0	32.9	60.6
33.0	63.0	33.1	65.0	33.2	66.5
33.3	67.6	33.4	68.1	33.5	68.2
33.6	67.6	33.7	66.5	33.8	64.8
33.9	62.7	34.0	60.0	34.1	56.9
34.2	53.5	34.3	49.9	34.4	46.1
34.5	42.3	34.6	38.5	34.7	34.9
34.8	31.6	34.9	28.6	35.0	26.0
35.1	23.8	35.2	22.1	35.3	20.8
35.4	20.0	35.5	19.5	35.6	19.3
35.7	19.4	35.8	19.6	35.9	19.8
36.0	20.0	36.1	20.1	36.2	20.0
36.3	19.6	36.4	19.0	36.5	18.1
36.6	16.9	36.7	15.4	36.8	13.7
36.9	11.9	37.0	10.0		

DO YOU WANT TO STORE INTERPOLATED FUNCTION? (Y OR N) N

PROGRAM COMPLETED

Figure 10. Listing of values of interpolated function for time-series sample problem.

PROGRAM LISTING FOR LANCZOS-PRECONDITIONED, FOLDED (SIN X)/X INTERPOLATOR

```
100 PRINT "L_LANCZOS-PRECONDITIONED FOLDED SINX/X INTERPOLATION"
110 INIT
120 DIM A$(55),C$(20),D$(5),G$(1),O$(24),U$(14)
130 DATA 10,120,10,80,1,1,1
140 READ C1,C2,D1,D2,Q1,Q2,Q3
150 C3=C2-C1
160 D3=D2-D1
170 REM ** ENTER INPUT DATA
180 GOSUB 410
190 REM ** ENTER NUMBER OF INTERPOLATION POINTS
200 GOSUB 950
210 REM ** DISPLAY ORIGINAL FUNCTION
220 GOSUB 1060
230 REM ** DETREND ORIGINAL FUNCTION
240 GOSUB 1200
250 REM ** CREATE AND APPEND ODD FUNCTION
260 GOSUB 1370
270 REM ** GENERATE COEFFICIENTS
280 GOSUB 1570
290 REM ** COMPUTE INTERPOLATED FUNCTION
300 GOSUB 1690
310 REM ** RESTORE TREND TO INTERPOLATED FUNCTION
320 GOSUB 1940
330 REM ** PLOT INTERPOLATED FUNCTION
340 GOSUB 2650
350 REM ** TABULATE INTERPOLATED FUNCTION
360 GOSUB 2810
370 REM ** STORE INTERPOLATED FUNCTION
380 GOSUB 3130
390 PRINT "G_G_G___PROGRAM COMPLETED"
400 END
410 REM ** SUB: ENTER INPUT DATA
420 PRINT "__DO YOU WANT TO INPUT VALUES FROM A TAPE? (Y OR N) ";
430 INPUT G$
440 IF G$="N" THEN 470
450 GOSUB 2570
460 RETURN
470 REM ** ENTER DATA FROM KEYBOARD
480 PRINT "__DO YOU WANT TO WORK WITH PROFILE DATA? (Y OR N) ";
490 INPUT G$
500 IF G$="Y" THEN 620
510 Q2=2
520 PRINT "__YOU HAVE ELECTED TO WORK WITH TIME-SERIES DATA"
530 PRINT "    Sample interval = ";
540 INPUT S1
550 PRINT "Units, 12 char max = ";
560 INPUT U$
570 PRINT "          Start time = ";
580 INPUT T1
590 PRINT "            End time = ";
600 INPUT T2
610 GO TO 700
620 PRINT "    Station spacing = ";
630 INPUT S1
640 PRINT "Units, 12 char max = ";
650 INPUT U$
660 PRINT "Start profile dist = ";
670 INPUT T1
680 PRINT "  End profile dist = ";
690 INPUT T2
700 N1=(T2-T1)/S1+1
710 PRINT "  Number of values = ";N1
720 PRINT "DATA IDENTIFICATION (max of 22 characters):
730 INPUT O$
740 DIM O(N1)
```

```
750 IF Q2=1 THEN 780
760 PRINT "DATA PT.    TIME       VALUE"
770 GO TO 790
780 PRINT "STA NO.   STA DIST   STA VALUE"
790 D=T1-S1
800 FOR J=1 TO N1
810 D=D+S1
820 PRINT "   ";J;"          ";D;"          ";
830 INPUT O(J)
840 NEXT J
850 PRINT "__DO YOU WANT TO STORE THESE VALUES ON A MDT? (Y OR N) ";
860 INPUT G$
870 IF G$="N" THEN 940
880 A$=O$
890 N2=N1
900 DIM Y2(N1)
910 Y2=0
920 GOSUB 3200
930 DELETE Y2
940 RETURN
950 REM ** SUB: ENTER NUMBER OF INTERPOLATION POINTS
960 PRINT "__NO. OF INTERPOLATION POINTS BETWEEN ADJACENT VALUES = ";
970 INPUT P1
980 P2=P1+1
990 S2=S1/P2
1000 N3=N1-1
1010 P3=PI/N3
1020 N8=2*N1-1
1030 N9=2*N3
1040 DIM L(N8),V(N1)
1050 RETURN
1060 REM ** SUB: DISPLAY ORIGINAL FUNCTION
1070 CALL "MIN",O,M3,I1
1080 CALL "MAX",O,M5,I2
1090 PRINT "__DO YOU WANT TO PLOT ORIGINAL FUNCTION? (Y OR N) ";
1100 INPUT G$
1110 IF G$="N" THEN 1190
1120 A$="   ORIGINAL FUNCTION"
1130 X2=N1
1140 DIM A(N1)
1150 A=O
1160 M1=M3
1170 M2=M5
1180 GOSUB 2160
1190 RETURN
1200 REM ** SUB: DETREND FUNCTION
1210 G3=O(N1)-O(1)
1220 G1=G3/(N1-1)
1230 G2=O(1)
1240 V=0
1250 FOR J=2 TO N1-1
1260 G2=G2+G1
1270 V(J)=O(J)-G2
1280 NEXT J
1290 PRINT "DO YOU WANT TO PLOT DETRENDED FUNCTION? (Y OR N) ";
1300 INPUT G$
1310 IF G$="N" THEN 1360
1320 A$="   DETRENDED ORIGINAL FUNCTION"
1330 DIM A(N1)
1340 A=V
1350 GOSUB 2130
1360 RETURN
1370 REM ** SUB: CREATE AND APPEND ODD FUNCTION
1380 FOR I=1 TO N1
1390 L(I)=V(I)
1400 NEXT I
1410 K=0
```

```
1420 FOR I=N1-1 TO 1 STEP -1
1430 K=K+1
1440 L(I+N1)=-V(K)
1450 NEXT I
1460 DELETE V
1470 PRINT "DO YOU WANT TO PLOT LANCZOS-PRECONDITIONED FUNCTION?";
1480 PRINT " (Y OR N) ";
1490 INPUT G$
1500 IF G$="N" THEN 1560
1510 A$="   FUNCTION AFTER LANCZOS PRECONDITIONING"
1520 DIM A(N8)
1530 A=L
1540 X2=N8
1550 GOSUB 2130
1560 RETURN
1570 REM ** SUB: GENERATE COEFFICIENTS
1580 DELETE C
1590 DIM C(N9,P1)
1600 FOR J=1 TO N9
1610 FOR K=1 TO P1
1620 P=K/P2
1630 M4=J-N1+1
1640 M=PI*(P-M4)
1650 C(J,K)=SIN(M)/(N9*TAN(M/N9))
1660 NEXT K
1670 NEXT J
1680 RETURN
1690 REM ** SUB: COMPUTE INTERPOLATED FUNCTION
1700 Q1=2
1710 N4=N3*(P1+1)+1
1720 DIM V(N4),V1(N3-1)
1730 V=0
1740 K=1
1750 FOR J=1 TO N3
1760 FOR P=1 TO P1
1770 K=K+1
1780 FOR M=1 TO N3-1
1790 K1=M-J+1+N3
1800 K2=-M-J+1+N3
1810 IF -M-J=>-(N3-1) THEN 1830
1820 K2=K2+2*N3
1830 IF K2<=N9 THEN 1850
1840 K2=K2-N9
1850 V1(M)=L(1+M)*(C(K1,P)-C(K2,P))
1860 NEXT M
1870 V(K)=SUM(V1)
1880 NEXT P
1890 K=K+1
1900 V(K)=L(J+1)
1910 NEXT J
1920 DELETE L
1930 RETURN
1940 REM ** SUB: RESTORE TREND TO INTERPOLATED FUNCTION
1950 DELETE W1
1960 G4=-G3/(N4-1)
1970 G5=-G4
1980 DIM W1(N4)
1990 FOR K=1 TO N4
2000 G5=G5+G4
2010 W1(K)=V(K)-G5
2020 NEXT K
2030 W1=O(1)+W1
2040 DELETE V
2050 RETURN
2060 REM ** SUB: DRAW BORDERS
2070 MOVE C1,D2
2080 RDRAW C3,0
```

```
2090 RDRAW 0,-D3
2100 RDRAW -C3,0
2110 RDRAW 0,D3
2120 RETURN
2130 REM ** SUB: QUICK-PLOT OF RESULTS
2140 CALL "MAX",A,M2,I2
2150 CALL "MIN",A,M1,I1
2160 IF Q1=1 THEN 2200
2170 I1=T1+(I1-1)*S2
2180 I2=T1+(I2-1)*S2
2190 GO TO 2220
2200 I1=T1+(I1-1)*S1
2210 I2=T1+(I2-1)*S1
2220 PRINT "L_";O$;"   ";A$
2230 PRINT "NO. OF VALUES=";X2;"    MIN=";M1;" @ ";I1;"    MAX";
2240 PRINT "=";M2;" @ ";I2
2250 IF Q2=1 THEN 2290
2260 PRINT "ORIGINAL TIME BEGINS AT ";T1;" AND ENDS AT ";T2;
2270 PRINT " WITH SAMP INT = ";S1;" ";U$
2280 GO TO 2310
2290 PRINT "ORIGINAL PROFILE STARTS AT ";T1;" AND ENDS AT ";T2;
2300 PRINT " WITH STA SPA = ";S1;" ";U$
2310 IF Q3=2 THEN 2510
2320 GOSUB 2060
2330 IF Q1=1 THEN 2360
2340 M1=M3
2350 M2=M5
2360 M4=0.1*(M2-M1)
2370 WINDOW 1,X2,M1-M4,M2+M4
2380 VIEWPORT C1,C2,D1,D2
2390 A2=SUM(A)
2400 A2=A2/X2
2410 MOVE 1,A2
2420 PRINT " MEAN"
2430 MOVE 1,A2
2440 RDRAW X2,0
2450 CALL "DISP",A
2460 DELETE A
2470 WINDOW 0,130,0,100
2480 VIEWPORT 0,130,0,100
2490 MOVE 0,0
2500 PRINT
2510 RETURN
2520 REM ** SUB: READY TO PROCEED
2530 PRINT "ARE YOU READY TO PROCEED? (Y OR N) ";
2540 INPUT G$
2550 IF G$="N" THEN 2530
2560 RETURN
2570 REM ** SUB: READ OBSERVED VALUES FROM DATA TAPE
2580 PRINT "G_G_G___INSERT MDT IN 4924    FILE NO. = ";
2590 INPUT F3
2600 FIND @2:F3
2610 READ @2:N1,O$,U$,Q2,S1,T1,T2
2620 DIM O(N1)
2630 READ @2:O
2640 RETURN
2650 REM ** SUB: PLOT INTERPOLATED FUNCTION
2660 PRINT "__DO YOU WANT TO PLOT INTERPOLATED FUNCTION? (Y OR N) ";
2670 INPUT G$
2680 IF G$="N" THEN 2800
2690 A$="INTERPOLATED FUNCTION WITH"
2700 P5=P1*N3
2710 D$=STR(P5)
2720 A$=A$&D$
2730 C$=" ADDITIONAL VALUES"
2740 A$=A$&C$
2750 DIM A(N4)
```

```
2760 A=W1
2770 X2=N4
2780 IF Q3=2 THEN 2800
2790 GOSUB 2130
2800 RETURN
2810 REM ** SUB: TABULATE INTERPOLATED FUNCTION
2820 PRINT "DO YOU WANT TO TABULATE INTERPOLATED FUNCTION? (Y OR N) ";
2830 INPUT G$
2840 IF G$="N" THEN 3120
2850 Q3=2
2860 GOSUB 2690
2870 GOSUB 2130
2880 DIM T(N4)
2890 T(1)=T1
2900 T3=T1
2910 FOR J=2 TO N4
2920 T3=T3+S2
2930 T(J)=T3
2940 NEXT J
2950 IMAGE 3(4D.D,3X,4D.D,3X)
2960 IF Q2=2 THEN 2990
2970 PRINT "STATION   VALUE   STATION   VALUE   STATION   VALUE"
2980 GO TO 3000
2990 PRINT "  TIME     VALUE     TIME     VALUE     TIME     VALUE"
3000 N5=N4-3*INT(N4/3)
3010 FOR J=1 TO N4-2 STEP 3
3020 PRINT USING 2950:T(J),A(J),T(J+1),A(J+1),T(J+2),A(J+2)
3030 NEXT J
3040 IF N5=0 THEN 3120
3050 IF N5=1 THEN 3070
3060 IF N5=2 THEN 3100
3070 IMAGE 4D.D,3X,4D.D
3080 PRINT USING 3070:T(J),A(J)
3090 GO TO 3120
3100 IMAGE 2(4D.D,3X,4D.D,3X)
3110 PRINT USING 3100:T(J),A(J),T(J+1),A(J+1)
3120 RETURN
3130 REM ** SUB: STORE INTERPOLATED FUNCTION
3140 PRINT "__DO YOU WANT TO STORE INTERPOLATED FUNCTION? (Y OR N) ";
3150 INPUT G$
3160 IF G$="N" THEN 3440
3170 N2=N4
3180 DIM Y2(N4)
3190 Y2=W1
3200 PRINT "G_G_G_INSERT MDT IN 4924    FILE NO. = ";
3210 INPUT F4
3220 L3=100+10*N2
3230 L3=INT(L3/256+1)*256
3240 PRINT "LENGTH OF FILE REQUIRED = ";L3
3250 PRINT "IS FILE LENGTH SUFFICIENT? (Y OR N) ";
3260 INPUT G$
3270 IF G$="Y" THEN 3340
3280 PRINT "G_G_G___INSERT MDT IN 4051"
3290 GOSUB 2520
3300 FIND F4
3310 MARK 1,L3
3320 PRINT "G_G_G___RETURN MDT TO 4924"
3330 GOSUB 2520
3340 FIND @2:F4
3350 WRITE @2:N2,A$,U$,Q2,S1,T1,T2,Y2
3360 PRINT @2,2:
3370 REM ** VERIFY WRITE/READ
3380 DELETE N2,A$,U$,Q2,S1,T1,T2,Y2
3390 FIND @2:F4
3400 READ @2:N2,A$,U$,Q2,S1,T1,T2
3410 DIM Y2(N2)
3420 READ @2:Y2
3430 PRINT "DATA ARE STORED AND RETRIEVABLE FROM FILE ";F4
3440 RETURN
```